
*                                                 VCC   ^
*                            VCC  ^           R1       /_\
*                                /_\       _/\  /\  /\__|
*                                 |      |    \/  \/
*                               __|_____/|\_______
*                              |         |        |
*                       R2      ->QP1    |  QP2 <-        R3
*          __________/\  /\  /\___`|_____|____|'____/\  /\  /\_
*         |            \/  \/    _'|   VPB    |`_     \/  \/   |
*         |  VNC1               |                |             |
*         |   _________________/|\_______________|      _______|
*         |  |                  |     ^     ^    |     |
*         |  |                  |    /_\   /_\   |     |
*         |  |     _____________|     |     |    |     |
*         |  |    | VNC2        |   <-      ->   |     |
*         |  |    |             | ||          || |     |
*         |  |    |             |_||MP1    MP2||_|     |
*         |  |    |             | ||_        _|| |     |
*         |  |    |             |    |      |    |     |
*         |_/|\__/|\_______     ->   |      |   <-     |
*            |    |        |      || |VG3   | ||       |
*            |    |        |   MP3||_|      |_||MP4    |
*            |    |        |     _|| |   VG4| ||_      |
*           _|    |_       |    |   _|_    _|_   |     |
*       __|'QN1  QN2`|_    | VG5|  / _ \  / _ \  |VG6  |
*      |  |`->    <-'| |   |    |  \/ \/  \/ \/  |     |
*  VIN |     |____|   _|_  ->   |  /\_/\  /\_/\  |   <-
*     _|_       |     ///    || |  \___/  \___/  | ||
*    /_  \  VNE_|_       MP5 ||_|   _|_    _|_   |_||MP6
*   // \  \   / _ \         _|| |   ///    ///   | ||_
*   \   \//   \/ \/I1      |    | I2          I3 |    |
*    \___/    /\_/\        |    |_______||_______|    |
*     _|_     \___/        |   _|_      ||      _|_   |
*     ///      _|_         |  / _ \     C0     / _ \  |
*              ///         |  \/ \/            \/ \/  |
*                          |  /\_/\            /\_/\  |
*                          |  \___/            \___/  |
*                          |   _|_ I4       I5  _|_   |
*                          |   ///              ///   | OUT2
*                    OUT1  |                          |  ____
*                          |_/\  /\  /\___/\  /\  /\__|_|    |  ___
*                          |   \/  \/  _|_  \/  \/      |    |_|OUT|
*                          |  R4       ///      R5      |BDIF| |___|
*                          |____________________________|    |
*                                                       |____|



.OPTIONS    GMIN = 1e-18
VCC         VCC   0    DC    5
I4          VG5   0    20u
I5          VG6   0    20u
MP5         OUT1  VG5  VS5   VCC  pchannel   w=10u l=1u
MP6         OUT2  VG6  VS6   VCC  pchannel   w=10u l=1u
QP1         VNC2  VPB  VCC        PNPL
QP2         VNC1  VPB  VCC        PNPL
R1          VCC   VPB  10K
R2          VPB   VS5  30K
R3          VPB   VS6  30K
MP1         VG3   VNC2 VCC   VCC  pchannel   w=10u l=1u
MP2         VG4   VNC1 VCC   VCC  pchannel   w=10u l=1u
MP3         VG5   VG3  VNC2  VCC  pchannel   w=10u l=1u
MP4         VG6   VG4  VNC1  VCC  pchannel   w=10u l=1u
I2          VG3   0    20u
I3          VG4   0    20u
R4          OUT1  0    10
R5          OUT2  0    10
C0          VG5   VG6  10p

QN1         VNC1  VIN  VNE       NPNV
QN2         VNC2  0    VNE       NPNV
I1          VNE   0    1n
VIN         VIN   0    AC        1m
BDIFF       OUT   0    V  =       V(OUT1) -V(OUT2)

.MODEL  PNPL    PNP( BF=50 VAF=20 )

.MODEL  NPNV    NPN( BF=50 VAF=20 )

.model  nchannel nmos (level=3)

.model  pchannel pmos (level=3)

.control
set pensize = 2

tran        1u   1m    0  1u
ac          dec      30       1M      100k
plot         db(out/vin)

.endc

.end

=====================END_OF_SPICE============================

Perhaps one of the most under appreciated aspects



to BiCMOS processes is what CMOS requirements do 
to bipolar performance. The lifetime of minority
carriers use to be in microsecs in the old bipolar
processes. Now life time is in milliseconds. 

What that means is that transistors can now operate
at much lower collector currents than before.

In addition, the gate resistance for CMOS transistors
is in effect that of very good glass. These two 
features were used to build the integrator below.  



*                                                 VCC   ^
*                            VCC  ^           R1       /_\
*                                /_\       _/\  /\  /\__|
*                                 |      |    \/  \/
*                               __|_____/|\_______
*                              |         |        |
*                       R2      ->QP1    |  QP2 <-        R3
*          __________/\  /\  /\___`|_____|____|'____/\  /\  /\_
*         |            \/  \/    _'|   VPB    |`_     \/  \/   |
*         |  VNC1               |                |             |
*         |   _________________/|\_______________|      _______|
*         |  |                  |     ^     ^    |     |
*         |  |                  |    /_\   /_\   |     |
*         |  |     _____________|     |     |    |     |
*         |  |    | VNC2        |   <-      ->   |     |
*         |  |    |             | ||          || |     |
*         |  |    |             |_||MP1    MP2||_|     |
*         |  |    |             | ||_        _|| |     |
*         |  |    |             |    |      |    |     |
*         |_/|\__/|\_______     ->   |      |   <-     |
*            |    |        |      || |VG3   | ||       |
*            |    |        |   MP3||_|      |_||MP4    |
*            |    |        |     _|| |   VG4| ||_      |
*           _|    |_       |    |   _|_    _|_   |     |
*       __|'QN1  QN2`|_    | VG5|  / _ \  / _ \  |VG6  |
*      |  |`->    <-'| |   |    |  \/ \/  \/ \/  |     |
*  VIN |     |____|   _|_  ->   |  /\_/\  /\_/\  |   <-
*     _|_       |     ///    || |  \___/  \___/  | ||
*    /_  \  VNE_|_       MP5 ||_|   _|_    _|_   |_||MP6
*   // \  \   / _ \         _|| |   ///    ///   | ||_
*   \   \//   \/ \/I1      |    | I2          I3 |    |
*    \___/    /\_/\  1nA   |    |_______||_______|    |
*     _|_     \___/        |   _|_      ||      _|_   |
*     ///      _|_         |  / _ \     C0     / _ \  |
*              ///         |  \/ \/    10pF    \/ \/  |
*                          |  /\_/\            /\_/\  |
*                          |  \___/            \___/  |
*                          |   _|_ I4       I5  _|_   |
*                          |   ///              ///   | OUT2
*                    OUT1  |                          |  ____
*                          |_/\  /\  /\___/\  /\  /\__|_|    |  ___
*                          |   \/  \/  _|_  \/  \/      |    |_|OUT|
*                          |  R4       ///      R5      |BDIF| |___|
*                          |____________________________|    |
*                                                       |____|

The goal is to create an integrator which has
unity gain in tens of milliHertz and has far
greater that unity gain at DC. 



And the other goal is to able to do it completely
on chip using standard processing with nothing
special.  


